Airyn differentiaaliyhtälö

Airyn differentiaaliyhtälö on hyvin yksinkertainen toisen kertaluvun lineaarinen ja homogeeninen differentiaaliyhtälö, joka kuitenkaan ei ole ratkaistavissa tavallisten alkeisfunktioiden avulla:

```
> airyyht:= diff(y(x), x, x)-x*y(x)=0;
```

$$airyyht := \left(\frac{d^2}{dx^2}\mathrm{y}(x)\right) - x\,\mathrm{y}(x) = 0$$

```
> ylrtk:= dsolve(airyyht, y(x));
```

$$ylrtk := \mathrm{y}(x) = \_C1\ \mathrm{AiryAi}(x) + \_C2\ \mathrm{AiryBi}(x)$$

Maple tuntee kuitenkin laajemman kokoelman funktioita, ja näiden avulla voidaan lausua sekä differentiaaliyhtälön yleinen ratkaisu että sen derivaatta.

```
> diff(ylrtk,x);
```

$$\frac{d}{dx}\mathrm{y}(x) = \_C1\ \mathrm{AiryAi}(1, x) + \_C2\ \mathrm{AiryBi}(1, x)$$

Kaksi lineaarisesti riippumatonta yksittäisratkaisua saadaan antamalla sopivat alkuehdot ja ratkaisemalla näistä vakiot. Aluksi yleinen ratkaisu määritellään funktioksi.

```
> airy:= unapply(rhs(ylrtk), x);
```

$$airy := x \to \_C1\ \mathrm{AiryAi}(x) + \_C2\ \mathrm{AiryBi}(x)$$

```
> vakiot:= solve({airy(0)=1, D(airy)(0)=0}, {_C1, _C2});
```

$$vakiot := \left\{\_C2 = \frac{1}{2}\Gamma\left(\frac{2}{3}\right)3^{(1/6)}, \_C1 = \frac{1}{2}\Gamma\left(\frac{2}{3}\right)3^{(2/3)}\right\}$$

```
> vakiot2:= solve({airy(0)=0, D(airy)(0)=1}, {_C1, _C2});
```

$$vakiot2 := \left\{\_C2 = \frac{1}{3}\frac{\pi\,3^{(1/3)}}{\Gamma\left(\frac{2}{3}\right)}, \_C1 = -\frac{1}{3}\frac{\pi\,3^{(5/6)}}{\Gamma\left(\frac{2}{3}\right)}\right\}$$

Saadut lausekkeet sisältävät uuden erikoisfunktion, gammafunktion. Tälle käytetään yleensä symbolia Γ (kreikkalainen kirjain iso gamma).

Vakioita vastaavat yksittäisratkaisut ovat

```
> rtk1:= subs(vakiot, airy(x));
```

$$rtk1 := \frac{1}{2}\Gamma\left(\frac{2}{3}\right)3^{(2/3)}\ \mathrm{AiryAi}(x) + \frac{1}{2}\Gamma\left(\frac{2}{3}\right)3^{(1/6)}\ \mathrm{AiryBi}(x)$$

```
> rtk2:= subs(vakiot2, airy(x));
```

$$rtk2 := -\frac{1}{3}\frac{\pi\, 3^{(5/6)}\, \mathrm{AiryAi}(x)}{\Gamma\left(\frac{2}{3}\right)} + \frac{1}{3}\frac{\pi\, 3^{(1/3)}\, \mathrm{AiryBi}(x)}{\Gamma\left(\frac{2}{3}\right)}$$

Näiden kuvaajista on nähtävissä eräitä toisen kertaluvun homogeeniyhtälölle luonteenomaisia piirteitä:

```
> plot({rtk1, rtk2}, x=-15..2);
```




```
>
```

Jos differentiaaliyhtälössä $y'' + ky = 0$ vakio k on positiivinen, kyseessä on vakiokertoiminen yhtälö, jonka ratkaisuna on $y = C_1 \sin(\sqrt{k}\,)\, x + C_2 \cos(\sqrt{k}\,)\, x$, ts. sini-kosini-värähtely. Värähtelyn taajuus on sitä suurempi, mitä suurempi k on. Negatiivisilla muuttujan x arvoilla Airyn yhtälö on tämäntyyppinen: Yhtälö voidaan kirjoittaa muotoon $y'' + |\,x\,|\, y = 0$, ja sen ratkaisuna näyttää olevan värähtely, jonka taajuus kasvaa, kun $|\,x\,|$ kasvaa.

Vastaavalla tavalla Airyn yhtälö voidaan rinnastaa positiivisilla muuttujan arvoilla yhtälöön $y'' - ky = 0$, missä k on positiivinen. Tämän ratkaisut muodostuvat eksponenttifunktioista: $y = C_1\, \mathbf{e}^{(\sqrt{k}\, x)} + C_2\, \mathbf{e}^{(-\sqrt{k}\, x)}$.

Kuvaajat näyttävät myös toisen kertaluvun homogeeniyhtälöiden ratkaisuille tyypillisen ominaisuuden: Jos kahdella lineaarisesti riippumattomalla ratkaisulla on nollakohtia, nämä vuorottelevat. Toisen ratkaisun kahden peräkkäisen nollakohdan välissä on täsmälleen yksi toisen ratkaisun nollakohta. Todistus perustuu Wronskin determinantin ominaisuuksiin.

Linkit

lineaarisen ja homogeenisen yhtälön ratkaisujoukko
Wronskin determinantti
algebrallinen ratkaiseminen Maplella (symalg.mws)
alkuehtoa vastaava yksittäisratkaisu Maplella (rtkalk.mws)
Airyn yhtälön numeerinen ratkaiseminen (numryh.mws)
Airyn yhtälön sarjaratkaisu

SKK & MS 31.05.2001